

Crear y publicar librerías

Descargar estos apuntes [pdf](#) o [html](#)

Migrar proyecto a Android Studio 2024 con Kotlin 2.0.24

1. File -> Project Structure -> Gradle Versión -> 8.7
2. Tools -> AGP upgrade assistant -> 8.6.0
3. Copiar `libs.versions.toml` a la carpeta `gradle`.
4. Cambiar el contenido de `settings.gradle.kts` por:

```
pluginManagement {
    repositories {
        google {
            content {
                includeGroupByRegex("com\\.\\.android.*")
                includeGroupByRegex("com\\.\\.google.*")
                includeGroupByRegex("androidx.*")
            }
        }
        mavenCentral()
        gradlePluginPortal()
    }
}
dependencyResolutionManagement {
    repositoriesMode.set(RepositoriesMode.FAIL_ON_PROJECT_REPOS)
    repositories {
        google()
        mavenCentral()
        maven {
            url = uri("https://jitpack.io")
        }
    }
}

rootProject.name = "NOMBRE DE MI PROYECTO"
include(":app")
```

5. En `build.gradle.kts`, **general a todos los módulos**, agregar los plugins:

```
plugins {
    alias(libs.plugins.android.application) apply false
    alias(libs.plugins.kotlin.android) apply false
    alias(libs.plugins.compose.compiler) apply false
    alias(libs.plugins.devtools.ksp) apply false
    alias(libs.plugins.com.google.dagger) apply false
}
```

6. En `build.gradle.kts`, de la aplicación:

- Agregar los plugins:

```
plugins {
    alias(libs.plugins.android.application)
    alias(libs.plugins.kotlin.android)
    alias(libs.plugins.compose.compiler)
    alias(libs.plugins.devtools.ksp)
    alias(libs.plugins.com.google.dagger)
    id("kotlin-parcelize")
}
```

- Eliminar bloque `composeOptions` dentro del bloque `android` ya que el compilador de compose viene ahora incluido en un plugin:

```
composeOptions {
    kotlinCompilerExtensionVersion = "1.5.0"
}
```

- En el bloque `dependencies` cambiar todas por:

```
dependencies {
    implementation(libs.androidx.core.ktx)
    implementation(libs.androidx.lifecycle.runtime.ktx)
    implementation(libs.androidx.activity.compose)
    implementation(platform(libs.androidx.compose.bom))
    implementation(libs.androidx.ui)
    implementation(libs.androidx.ui.graphics)
    implementation(libs.androidx.ui.tooling.preview)
    implementation(libs.androidx.material3)

    implementation(libs.pmdm.ies.balmis)

    implementation(libs.dagger.hilt.android)
    implementation(libs.androidx.hilt.navigation.compose)
    ksp(libs.dagger.hilt.android.compiler)
    kspAndroidTest(libs.dagger.hilt.android.compiler)

    implementation(libs.compose.navigation)

    implementation(libs.androidx.room.ktx)
    ksp(libs.androidx.room.compiler)

    testImplementation(libs.junit)
    androidTestImplementation(libs.androidx.junit)
    androidTestImplementation(libs.androidx.espresso.core)
    androidTestImplementation(platform(libs.androidx.compose.bom))
    androidTestImplementation(libs.androidx.ui.test.junit4)
    debugImplementation(libs.androidx.ui.tooling)
    debugImplementation(libs.androidx.ui.test.manifest)
}
```

7. Hacer **gradle sync** y hacer el **Scan for deprecations** que propone al migrar la versión de Kotlin.
8. En el archivo `ui.theme.Theme.kt` eliminar el bloque deprecado y **optimizar imports**.

```
...
val view = LocalView.current
if (!view.isInEditMode) {
    SideEffect {
        (view.context as Activity).window.statusBarColor = colorScheme.primary.toArgb()
        ViewCompat.getWindowInsetsController(view)?.isAppearanceLightStatusBars = darkTheme.isLight
    }
}
...
```

9. Eliminar paquetes `ui.composables`, `utilities.validacion` y `manejo_errores`. Posteriormente arreglar dependencias con la librería nueva.
 - **Replace in Files (Ctrl+Shift+H)** todos los imports de imports en el proyecto de `com.pmdm.XXXX.utilities.validacion` y sustituirlos por

```
com.github.pmdmiesbalmis.utilities .
```

- Si hay errores de iconos ir a <https://fonts.google.com/icons> :
 - Marcar el switch fill.
 - Buscar el icono deseado y seleccionar la carpeta Android.
 - Descargar el recurso xml desde la web e incluirlo en la carpeta `res/drawable` y sustituirlo por el correspondiente

```
import androidx.compose.material.icons.filled.XXXX .
```

10. Cuando compile todo. Botón derecho sobre App -> Refactor -> **Migrate Build Config to Gradle Build Files**. Pulsar **Do Refactor**.